
Drax Transport Model
This raku workbook is intended to help understand the claims made by the Drax power station in the UK

Drax is converting from burning coal to sustainable wood pellets. The level to which this is truly contributing to global CO2

reduction is a contraversial topic.

More information can be found at https://www.clientearth.org/projects/the-greenwashing-files/drax/

0. Some Setup

1. A Load of Pellets

The Zheng Zhi bulk carrier vessel transported 63907metric-ton of Drax’s sustainable biomass from the 
US port of Baton Rouge in Louisiana.

Woody biomass has an average heating density of 19.8MJ/kg.

Each carrier load delivers 1.27PJ.

2. Transport Efficiency

The Zheng Zhi's 8000km round trip from Baton Rouge to Yorkshire is covered at a speed of 24knot 
which takes about 7.5day.

A Panamax ship can consume 63000gallons per day of marine fuel at that speed, giving a total of 1788.46m^3 
for the round trip.

The density of marine oil is 860kg/m³, so the total mass burnt per trip is 1538.08metric-ton.

The approximate fraction of fuel to load is 2.41%.

3. Annual Total

In 2020 Drax’s reported annual emissions from burning biomass totaled 13.2 million tonnes.

In 2020 Drax’s reported annual emissions from burning biomass totaled 13.2million tonnes.

This required 206.55① round trips of a Panamax class bulk carrier.

Since each round trip takes 7.5day, then that means each vessel can do 48.7① per year.

And that requires 5 vessels continually plying the ocean.

In [1]:
use Physics::Unit;
use Physics::Measure :ALL;

$Physics::Measure::round-val = 0.01;

Out[1]:

In [2]:
#First, we will need to define some new units...

#| define the Units:
Unit.new(defn => 'm^3 / s', names => ['m^3/s']);
Unit.new(defn => 'us-gallons per day', names => ['gallons per day']);

#| declare a new Measure type:
class ConsumptionRate is Measure {
    has $.units where *.name eq <m^3/s gpd>.any;
}

#| link the base type to the units
GetMeaUnit('m^3/s').NewType('ConsumptionRate');

say "ok"

In [3]:
my $zz-load = 

♎

'63907 tonnes';

say "The Zheng Zhi bulk carrier vessel transported $zz-load of Drax’s sustainable biomass from the 
US port of Baton Rouge in Louisiana."

In [4]:
my $energy-per-kg = 

♎

'19.8 MJ/kg';

say "Woody biomass has an average heating density of $energy-per-kg."

In [5]:
my $zz-energy = ( $energy-per-kg * $zz-load ).norm;

say "Each carrier load delivers $zz-energy."

In [16]:
my $trip-dist = 8000km;
my $trip-speed = 

♎

'24 knots';
my $trip-time = ( $trip-dist / $trip-speed ).in: <days>;

say "The Zheng Zhi's $trip-dist round trip from Baton Rouge to Yorkshire is covered at a speed of $trip-speed 
which takes about $trip-time.";

In [8]:
my $fuel-day = 

♎

'63_000 gallons per day';
my $fuel-total = $fuel-day * $trip-time;

say "A Panamax ship can consume $fuel-day of marine fuel at that speed, giving a total of $fuel-total 
for the round trip."

In [11]:
my $fuel-density = 

♎

'860 kg/m³';
my $fuel-mass = ($fuel-total * $fuel-density).in: <tonnes>;

say "The density of marine oil is $fuel-density, so the total mass burnt per trip is $fuel-mass."

In [13]:
my $fuel-fraction = ( $fuel-mass / $zz-load ).in: <%>;

say "The approximate fraction of fuel to load is $fuel-fraction."

In [14]:
my $annual-mass = 

♎

'13.2 million tonnes';

say "In 2020 Drax’s reported annual emissions from burning biomass totaled $annual-mass."

In [31]:
my $annual-trips = $annual-mass / $zz-load;

say "This required $annual-trips round trips of a Panamax class bulk carrier."

In [30]:
my $zz-annual-trips = 

♎

'1 year' / $trip-time;

say "Since each round trip takes $trip-time, then that means each vessel can do $zz-annual-trips per year."

In [38]:
say "And that requires {(+$annual-trips / +$zz-annual-trips).ceiling} vessels continually plying the ocean."

In [ ]:

